
Overview

This document lists a TTCN change needed to correct a problem in ATS ‘iwd-EUTRA-B2011-03_D11wk27’.

Corrections 

Change 1 – Correction to UTRAN PS RB Establishment 

Function name
f_UTRAN_UE_PsRB_Est()

Reason for change
In line with previously agreed changes in R5s110315 and R5s110386, the ‘GMM Service Request’ message and the ‘Activate PDP Context Accept’ message should be sent over ‘utran_CellDedicated’ instead of the specific identity  ‘p_CellId’ (= Cell 5).

Summary of change
‘utran_CellDedicated’ instead of ‘p_CellId’ is used as parameter’ in functions f_UTRAN_GMM_ServReq , f_UTRAN_PS_SessionAct_ReceivePDP_Req  and f_UTRAN_SendActivatePDP_Accept. 

TTCN module
UTRAN_RAB_Functions.ttcn

MCC160 Comment


Before change

  //------------------------------------------------------------------------------------
  // Radio Bearer Establishement in UTRAN
  //------------------------------------------------------------------------------------
  /*
   * @desc      Performs UTRA PS RB Establishment according to TS 34.108 subclause 7.4.2 in a UTRAN cell.
   * @param     p_CellId
   * @param     p_RB_Configuration .. Radio Bearer type to be configured
   * @status    APPROVED
   */
  function f_UTRAN_UE_PsRB_Est ( UTRAN_CellId_Type p_CellId,
                                 RB_ConfigType     p_RB_Configuration ) runs on UTRAN_PTC
  {
    var template (value) EstablishmentCause v_EstCause := fl_GetMoEstCause();
    var RB_ConfigType v_CellConfiguration := f_UTRAN_CellInfo_GetConfigType ( p_CellId );
    var ACTIVATEPDPCONTEXTREQUESTul v_ActPdpContextReq;
    var template (value) PktDataProtoAddr v_ExpectedPDN_Address;
    var boolean v_NewKeyFlag := true;
    var TI v_Ti_R, v_Ti_S;// Transaction Identifier. r = value used to receive CC PDU. s = value used to send CC PDU.
    var template (value) QualityOfService v_QualityOfService;
    var integer v_DedicatedId := 1;
    //Init QoS and PDN Address parameters only if PS RAB is to be configured
    if ( (p_RB_Configuration == cell_DCH_64kPS_RAB_SRB) or
         (p_RB_Configuration == cell_DCH_HS_DSCH) or
         (p_RB_Configuration == cell_DCH_E_HS) or
         (p_RB_Configuration == cell_FACH_PS))
      {
        v_ExpectedPDN_Address := f_UG_GetPktDataProtoAddr (true);
        v_QualityOfService := f_UTRAN_InitialiseQoS (p_CellId, p_RB_Configuration);
      }
    else
      {
        FatalError(__FILE__, __LINE__, "Configuration not supported");
      }
    //Initiate an outgoing PS call
    f_UT_ActivatePDPContextUG (UT, v_DedicatedId, v_QualityOfService, "");
    if (( v_CellConfiguration == cell_DCH_StandAloneSRB_NoConn) or
        (v_CellConfiguration == cell_FACH_NoConn))
      {
        //Steps 1-3 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
        f_UTRAN_RRC_ConnEst (p_CellId, v_EstCause);
        // Step 4 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
        f_UTRAN_GMM_ServReq (p_CellId, ps_domain);
        //Perform NAS session activation procedure for packet switched sessions (34.108 cl. 7.4.2.4.2)
        //Steps 1-2 in 34.108 cl. 7.4.2.4.2: Generate authentication and run authentication procedure
        f_UTRAN_GMM_Authentication();
        //Step 3-4 in 34.108 cl. 7.4.2.4.2: Transmit Security Mode Comamnd & Receive Security Mode Comlpete
        f_UTRAN_RRC_Security (p_CellId, v_NewKeyFlag, ps_domain);
        //Step 5 in 34.108 cl. 7.4.2.4.2: Receive PDP Context Activation Request
        v_ActPdpContextReq := f_UTRAN_PS_SessionAct_ReceivePDP_Req (p_CellId, p_RB_Configuration, v_QualityOfService);
        //Assign v_RAB_Id (BITSTRING[8]) with the NSAPI (BITSTRING[4]) value received in ACTIVATE PDP CONTEXT REQUEST message:
        //v_RAB_Id := int2bit (bit2int (v_ActPdpContextReq.requestedNSAPI.nSAPI_Value), 8);
        //Radio Bearer Identity is set according to 34.108 cl 9.1.1 RADIO BEARER SETUP message for condition A3, A6 or A9.
        v_Ti_R := v_ActPdpContextReq.ti;
        v_Ti_S.tiFlag := '1'B;
        v_Ti_S.tiVal := v_Ti_R.tiVal;
        //Step 1-3 in 34.108 cl. 7.4.2.6.2.3: Radio access bearer establishment procedure for PS mobilie originated calls.
        //Step 1-2 in 34.108 cl. 7.4.2.6.2.3: Send RADIO BEARER SETUP and receive COMPLETE
        f_UTRAN_RB_SetUp (p_CellId, p_RB_Configuration);
        //Step 3 in 34.108 cl. 7.4.2.6.2.3: Send ACTIVATE PDP CONTEXT ACCEPT
        f_UTRAN_SendActivatePDP_Accept (p_CellId,
                                        v_ExpectedPDN_Address.pDP_TypeNo,
                                        v_ExpectedPDN_Address.addrInfo,
                                        v_ExpectedPDN_Address.iel,
                                        v_Ti_S,
                                        v_QualityOfService,
                                        p_RB_Configuration);
      }
    else
      {
        FatalError(__FILE__, __LINE__, "Cell is not properly configured");
      }
  }

After change

  //------------------------------------------------------------------------------------
  // Radio Bearer Establishement in UTRAN
  //------------------------------------------------------------------------------------
  /*
   * @desc      Performs UTRA PS RB Establishment according to TS 34.108 subclause 7.4.2 in a UTRAN cell.
   * @param     p_CellId
   * @param     p_RB_Configuration .. Radio Bearer type to be configured
   * @status    APPROVED
   */
  function f_UTRAN_UE_PsRB_Est ( UTRAN_CellId_Type p_CellId,
                                 RB_ConfigType     p_RB_Configuration ) runs on UTRAN_PTC
  {
    var template (value) EstablishmentCause v_EstCause := fl_GetMoEstCause();
    var RB_ConfigType v_CellConfiguration := f_UTRAN_CellInfo_GetConfigType ( p_CellId );
    var ACTIVATEPDPCONTEXTREQUESTul v_ActPdpContextReq;
    var template (value) PktDataProtoAddr v_ExpectedPDN_Address;
    var boolean v_NewKeyFlag := true;
    var TI v_Ti_R, v_Ti_S;// Transaction Identifier. r = value used to receive CC PDU. s = value used to send CC PDU.
    var template (value) QualityOfService v_QualityOfService;
    var integer v_DedicatedId := 1;
    //Init QoS and PDN Address parameters only if PS RAB is to be configured
    if ( (p_RB_Configuration == cell_DCH_64kPS_RAB_SRB) or
         (p_RB_Configuration == cell_DCH_HS_DSCH) or
         (p_RB_Configuration == cell_DCH_E_HS) or
         (p_RB_Configuration == cell_FACH_PS))
      {
        v_ExpectedPDN_Address := f_UG_GetPktDataProtoAddr (true);
        v_QualityOfService := f_UTRAN_InitialiseQoS (p_CellId, p_RB_Configuration);
      }
    else
      {
        FatalError(__FILE__, __LINE__, "Configuration not supported");
      }
    //Initiate an outgoing PS call
    f_UT_ActivatePDPContextUG (UT, v_DedicatedId, v_QualityOfService, "");
    if (( v_CellConfiguration == cell_DCH_StandAloneSRB_NoConn) or
        (v_CellConfiguration == cell_FACH_NoConn))
      {
        //Steps 1-3 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
        f_UTRAN_RRC_ConnEst (p_CellId, v_EstCause);
        // Step 4 in 34.108 cl. 7.4.2.2.2.3: RRC connection establishment procedure for MO PS calls on DCH
        f_UTRAN_GMM_ServReq (utran_CellDedicated, ps_domain);
        //Perform NAS session activation procedure for packet switched sessions (34.108 cl. 7.4.2.4.2)
        //Steps 1-2 in 34.108 cl. 7.4.2.4.2: Generate authentication and run authentication procedure
        f_UTRAN_GMM_Authentication();
        //Step 3-4 in 34.108 cl. 7.4.2.4.2: Transmit Security Mode Comamnd & Receive Security Mode Comlpete
        f_UTRAN_RRC_Security (p_CellId, v_NewKeyFlag, ps_domain);
        //Step 5 in 34.108 cl. 7.4.2.4.2: Receive PDP Context Activation Request
        v_ActPdpContextReq := f_UTRAN_PS_SessionAct_ReceivePDP_Req (utran_CellDedicated, p_RB_Configuration, v_QualityOfService);
        //Assign v_RAB_Id (BITSTRING[8]) with the NSAPI (BITSTRING[4]) value received in ACTIVATE PDP CONTEXT REQUEST message:
        //v_RAB_Id := int2bit (bit2int (v_ActPdpContextReq.requestedNSAPI.nSAPI_Value), 8);
        //Radio Bearer Identity is set according to 34.108 cl 9.1.1 RADIO BEARER SETUP message for condition A3, A6 or A9.
        v_Ti_R := v_ActPdpContextReq.ti;
        v_Ti_S.tiFlag := '1'B;
        v_Ti_S.tiVal := v_Ti_R.tiVal;
        //Step 1-3 in 34.108 cl. 7.4.2.6.2.3: Radio access bearer establishment procedure for PS mobilie originated calls.
        //Step 1-2 in 34.108 cl. 7.4.2.6.2.3: Send RADIO BEARER SETUP and receive COMPLETE
        f_UTRAN_RB_SetUp (p_CellId, p_RB_Configuration);
        //Step 3 in 34.108 cl. 7.4.2.6.2.3: Send ACTIVATE PDP CONTEXT ACCEPT
        f_UTRAN_SendActivatePDP_Accept (utran_CellDedicated,
                                        v_ExpectedPDN_Address.pDP_TypeNo,
                                        v_ExpectedPDN_Address.addrInfo,
                                        v_ExpectedPDN_Address.iel,
                                        v_Ti_S,
                                        v_QualityOfService,
                                        p_RB_Configuration);
      }
    else
      {
        FatalError(__FILE__, __LINE__, "Cell is not properly configured");
      }
  }

